
Softkey
Executing a Procedure with a Softkey (User Menu Function) 
The E5070B/E5071B lets you perform procedures assigned to specific softkeys ([Macro 
Setup] - User Menu - Button 1/2/3/4/5/6/7/8/9/10) without using user forms by an event 
that the softkey is pressed. This function is called the user menu function. 

NOTE You do not have to execute any VBA program when using the user menu function. 

Preparation for Using the User Menu Function 

Before using the user menu function, perform the following preparation. 

Coding of a Procedure Assigned to a Softkey 

Follow these steps to create a procedure assigned to a specific softkey in the "UserMenu" 
object in the "E5070 Objects" folder. 

Step 1. Double-click the "UserMenu" icon in the "E5070 Objects" folder to open the code window. 

Step 2. In the object box in the code window, click UserMenu as shown below. 

Step 3. In the UserMenu_OnPress(ByVal id As Long) procedure, create a program you want 
to assign to a specific softkey (specify with the id variable). For actual use example, see 
Line 70 to 430 in the Example 1-2. 

NOTE During processing an event (during execution of a procedure for a key pressed), another 
event (an interrupt by a procedure for another softkey pressed) cannot be accepted. 

NOTE You cannot save (export) the "UserMenu" object by module basis; save it by project basis. 



Settings for Softkey Label and Softkey Enabled/Disenabled

When you want to change the softkey labels for the user menu function, use the following 
COM object. For more information on this object, see Chapter 7, “COM Object Reference” 
in E5070B/E5071B VBA Programmer’s Guide.

• UserMenu.Item(id).Caption

When you want to set the softkey enabled/disenabled for the user menu function, use the 
following COM object. For more information on this object, see Chapter 7, “COM Object 
Reference” in E5070B/E5071B VBA Programmer’s Guide. 

• UserMenu.Item(id).Enabled

Moreover, when you want to preset the above settings for the user menu function, use the 
following COM object. For more information on this object, see Chapter 7, “COM Object 
Reference” in E5070B/E5071B VBA Programmer’s Guide. 

• UserMenu.PRESet 

NOTE The above user menu setting is also presetted by pressing [Macro Setup] - Preset User 
Menu on the E5070B/E5071B front panel.

How to Use the User Menu Function 

To execute a procedure assigned to a softkey, you need to generate an event of pressing the 
softkey. To generate an event, the manual method and the COM object method are 
available. 

Method by Manual Operation

Step 1. Click the specific softkey as follows: 

• [Macro Setup] - User Menu - Button No.

"No." represents a button number. You can set the label for "Button No." as you like. 
For detail, refer to the “Settings for Softkey Label and Softkey Enabled/Disenabled.” 
section. 

Method by COM Object 

You can use the following COM object to perform the same operation as pressing a 
specific softkey. For more information on this object, see Chapter 7, “COM Object 
Reference” in E5070B/E5071B VBA Programmer’s Guide. 

• UserMenu.Press(id) 

Simple Example 

The sample program disk contains a sample program, named “meas_user.vba“, that 
demonstrates how to use the user menu function. This VBA program consists of the 
following standard module and the "UserMenu" object. 

The program (object name: mdlUserMenu) is described in detail below:

Line 70  Stores True into the State variable. 

Lines 90 to 150  Sets the first to third softkey (id: 1 to 3) enabled, and sets the fourth to 
tenth softkey (id: 4 to 10) disenabled. 

Lines 170 to 190  Sets the first softkey label (id: 1) to “Setup” the second softkey label 
(id: 2) to “Meas” the third softkey label (id: 3) to “Exit”. 

Line 210  Displays the buttons for the user menu function in the softkey area. 

Lines 230 to 250  Processing repeated until the State variable is True (State = True). 

Line 240: Detects an event that a specific softkey is pressed and 
enables the interrupt from the event. 

Example 1-1 Sample program using user menu (object name: mdlUserMenu) 

  10|  Public State As Boolean
  20|  
  30|  Sub Main()
  40|    
  50|    Dim I As Long, J As Long
  60|    
  70|    State = True
  80|    
  90|    For I = 1 To 3
 100|      UserMenu.Item(I).Enabled = True
 110|    Next I
 120|    
 130|    For J = 4 To 10
 140|      UserMenu.Item(J).Enabled = False
 150|    Next J
 160|    
 170|    UserMenu.Item(1).Caption = "Setup"
 180|    UserMenu.Item(2).Caption = "Meas"
 190|    UserMenu.Item(3).Caption = "Exit"
 200|    
 210|    UserMenu.Show
 220|    
 230|    Do While State
 240|      DoEvents
 250|    Loop
 260|  
 270|  End Sub

Object name Module type Content

mdlUserMenu Standard module Sets the softkey labels and enables interrupts 
from the softkeys. 



The procedures of the "UserMenu" object are described below. 

Lines 70 to 190  The procedure when the first softkey (id: 1) is pressed. 

Line 90: Returns the E5070B/E5071B to the preset state. 

Lines 110 to 130 For channel 1, sets the sweep start value to 1.73 
GHz, the sweep stop value to 1.83 GHz, and the number of 
measurement points to 51. 

Lines 150 to 170 After aborting the measurement, sets the trigger 
source to the bus trigger and turns on the continuous trigger startup 
mode for channel 1. 

Line 190: Displays the buttons for the user menu function in the 
softkey area. 

Lines 210 to 320  The procedure when the second softkey (id: 2) is pressed. 

Lines 230 to 240 Generates a trigger to start a single sweep and 
waits until the measurement finishes (1 is read out with the 
SCPI.IEEE4882.OPC object). 

Line 260: Retrieves the number of points in channel 1 and stores 
that number into the Nop variable. 

Lines 280 to 290 Specifies trace 1 of channel 1 to the active trace, 
retrieves the formatted data array, and stores the data into the 
FmtData variable. 

Lines 310 to 320 Displays the echo window in the lower part of the 
LCD screen. 

Lines 340 to 360: Displays 2 measurement data values (primary 
value and secondary value) for each measurement point in the echo 
window. 

Lines 380 to 430  The procedure when the third softkey (id: 3) is pressed. 

Line 400: Displays a program closing message. 

Line 410: Stores False into the sta variable to terminate the main 
program. 

Example 1-2 Sample program using user menu ("UserMenu" object) 

  10|  Private Sub UserMenu_OnPress(ByVal id As Long)
  20|  
  30|  Dim I As Integer
  40|  Dim Nop As Long, Dmy As Long
  50|  Dim FmtData As Variant
  60|  
  70|  If id = 1 Then
  80|  
  90|      SCPI.SYSTem.PRESet
 100|      
 110|      SCPI.SENSe(1).FREQuency.STARt = 1730000000#
 120|      SCPI.SENSe(1).FREQuency.STOP = 1830000000#
 130|      SCPI.SENSe(1).SWEep.POINts = 51
 140|      
 150|      SCPI.ABORt

 160|      SCPI.TRIGger.SEQuence.Source = "BUS"
 170|      SCPI.INITiate(1).CONTinuous = True
 180|      
 190|      UserMenu.Show
 200|  
 210|  ElseIf id = 2 Then
 220|      
 230|      SCPI.TRIGger.SEQuence.SINGle
 240|      Dmy = SCPI.IEEE4882.OPC
 250|      
 260|      Nop = SCPI.SENSe(1).SWEep.POINts
 270|      
 280|      SCPI.CALCulate(1).PARameter(1).SELect
 290|      FmtData = SCPI.CALCulate(1).SELected.DATA.FDATa
 300|      
 310|      SCPI.DISPlay.TABLe.TYPE = "ECHO"
 320|      SCPI.DISPlay.TABLe.State = True
 330|          
 340|      For I = 1 To Nop - 1
 350|      ECHO FmtData(2 * I - 2), FmtData(2 * I - 1)
 360|      Next I
 370|  
 380|  ElseIf id = 3 Then
 390|      
 400|      MsgBox "Program ended!"
 410|      State = False
 420|  
 430|  End If
 440|  
 450|  End Sub

